* regressions_informal_edom_ebite_industry.do	JEP	03/11/15
* Does regressions that are in the term paper, for informal workers: RIF10
* Exclude domestic workers
* Monthly wage
* Extended bite

*************************** Change log ********************************** 

/*
	
*/	

clear 
version 13.1

* Project info
cap project, doinfo
if _rc==198 {
	if c(os)=="Unix" loc master "/home/jperez/AA_Minimum_Wage" 
else if c(username)=="J16339" | c(username)=="A15858" loc master "B:\Col_Minimum_Wage" 
 	else loc master "C:/Users/jorpp/Dropbox (Brown)/Col_Minimum_Wage"
 loc pr = 0
}
else {
	local master "`r(pdir)'"
	local doname "`r(dofile)'"  
 loc pr = 1
	* Project calls
	project, uses("`master'/Data/Source/ocupados_mw.dta")
	* Call programs
	project, relies_on("`master'/Code/Est/sample_choice.do")
	project, relies_on("`master'/Code/Est/build_mw_measures.do")
	project, relies_on("`master'/Code/Est/build_treatment.do")
	project, relies_on("`master'/Code/Est/export_rif.do")
	project, relies_on("`master'/Code/Est/confirmedrun.do")
	project, relies_on("`master'/Code/Est/graph_rif.do")
}
*/


* Call necessary programs. I do not call using project to carry the programs around. I do not put these in the ado/personal folder because they are specific to the project. They are not ado files.
do "`master'/Code/Est/sample_choice.do"
do "`master'/Code/Est/build_mw_measures.do"
do "`master'/Code/Est/build_treatment.do"
do "`master'/Code/Est/export_rif.do"
do "`master'/Code/Est/confirmedrun_cities.do"
do "`master'/Code/Est/graph_rif_cities.do"

use "`master'/Data/Source/ocupados_mw.dta", clear

* Choose quantiles
glo qtiles "5 10 15 20 25 30 35 40 45 50 55 60 65 70 75 80 85 90"
* Number of bootstraps
glo boot=200

***

* RIF 4 - Tables numbered 4 in term paper

* Choose sample
mysample rif10sample , hmin(30) hmax(50) sex(0) edadm(65) wage(salario) informal exind(13,29,42)
tab rif10sample
tab tipo_trabajador if rif10sample
tab year if rif10sample
tab horas_semana if rif10sample 
tab sex if rif10sample
tab afiliado_salud if rif10sample
* Build mw measures
mw_measures rif10sample , stub(rif10)
* Build treatment
mytreatment, posts("tq(1999q1)") mw(bite_e_rif10_ind) wage(salario_mensual_real)

* RIF regressions

* Create estimation results folder if it doesn't already exist
cap mkdir "`master'/Ster/rif10_new_ebite_indcc_cities"

* For these I'm going to assume the clusters are the city - industry pairs. This assumes that there is not correlation within city across industries, only within those pairs. Could also run with the standard clustering by city, but that would mean that I'm not really increasing the sample size by running within industry.

* The number of clusters is large enough that I don't need the full CGM framework, can run just with cluster
egen city_ind=group(ccode cactividad_empresa)

cap program drop mex
program define mex
	cap mkdir "`2'/Doc/ster_figures/rif10_new_ebite_indcc_cities"
	cap mkdir "`2'/Doc/ster_figures/rif10_new_ebite_indcc_cities/1999q1"
	graph export "`2'/Doc/ster_figures/rif10_new_ebite_indcc_cities/1999q1/c`3'_`1'.pdf", replace
	*! epstopdf "`2'/Doc/ster_figures/rif10_new_ebite_indcc_cities/1999q1/`1'.eps"
	*erase "`2'/Doc/ster_figures/rif10_new_ebite_indcc_cities/1999q1/`1'.eps"
	* project, creates("`master'/Doc/descriptive_figures/density/`1'.pdf")
end	

levelsof ccode , loc(cities)
foreach c in `cities'{
preserve
keep if ccode == `c'
	foreach x in $qtiles {

		di " Percentile `x'"
		loc fw: word 1 of $qtiles
		loc count: word count $qtiles
		di `count'
		loc lw: word `count' of $qtiles
		if `x'==`lw' & `pr'!=0 loc reg="register"
		else loc reg ""
		if `x'==`fw' loc r="replace"
		else loc r="append"
	
		
		* Panel B. Employment and Bartik Price Variable
		confirmedrun, `reg' controls() name(`master'/Ster/rif10_new_ebite_indcc_cities/c`c'_reg_emp_bp_tq(1999q1)_ebite_salario_mensual_real_50_edom_indcc) rif(rif10B) p(`x') r(`r') sample(rif10sample) fe(i.cactividad_empresa i.time) cluster(cactividad) rerun
		* Panel E. Employment, Bartik Price Variable and City Trends
		confirmedrun, `reg' controls(i.cactividad#c.trend ) name(`master'/Ster/rif10_new_ebite_indcc_cities/c`c'_reg_sst_emp_bp_tq(1999q1)_ebite_salario_mensual_real_50_edom_indcc) rif(rif10E) p(`x') r(`r')  fe(i.cactividad_empresa i.time) cluster(cactividad)sample(rif10sample)  rerun
		* Panel F. Employment, Bartik Quantity Variable and City Trends
		* Panel N. Employment, Bartik Price Variable and national employment by industry
		confirmedrun, `reg' controls(empind ) name(`master'/Ster/rif10_new_ebite_indcc_cities/c`c'_reg_ie_emp_bp_tq(1999q1)_ebite_salario_mensual_real_50_edom_indcc) rif(rif10N) p(`x') r(`r') sample(rif10sample) fe(i.cactividad_empresa i.time) cluster(cactividad) rerun
		* Panel O. Employment, Bartik Price Variable and national employment by industry, city trends
		confirmedrun, `reg' controls(empind  i.cactividad#c.trend) name(`master'/Ster/rif10_new_ebite_indcc_cities/c`c'_reg_iect_emp_bp_tq(1999q1)_ebite_salario_mensual_real_50_edom_indcc) rif(rif10O) p(`x') r(`r') sample(rif10sample) fe(i.cactividad_empresa i.time) cluster(cactividad) rerun
		
	}

* Graph results and export

	clear
	*foreach spec in reg_emp reg_emp_bp reg_emp_bq reg_sst_emp_bp reg_sst_emp_bq reg_emp_bq_ixc reg_emp_bp_ixc reg_ct_emp_bp_ixc reg_ct_emp_bq_ixc	{
	* foreach spec in reg_emp_bp reg_emp_bq reg_sst_emp_bp reg_sst_emp_bq reg_emp_bq_ixc reg_emp_bp_ixc {
	foreach spec in reg_emp_bp reg_sst_emp_bp reg_ie_emp_bp reg_iect_emp_bp {
		graph drop _all
		graphrif, master("`master'") ster(Ster) folder(rif10_new_ebite_indcc_cities) qtiles($qtiles) spec(`spec') time(tq(1999q1)) city(`c') rareaopts(bcolor(gs13))twowayopts(title(Informal sector - ebite - By city industry - `spec' ) xtitle(Percentile) ytitle(Coef)) bite(ebite)
		mex `spec' `master' `c'
	}

restore
}

